§2.4 Depth-integrated hydrodynamical models

This is a Wolfram Mathematica script, following the mathematical derivations of the two depth-integrated hydrodynamical models in §2.4. It should be executed in the Wolfram Mathematica environment. A corresponding .pdf file, presenting the outputs of the programme, is also provided.

§2.4.1 Revised lubrication model

Semi-parabolic profile

```
In[1]:= F0[zeta_] := zeta - 1 / 2 zeta^2
```

Velocity wall-normal profile ansatz

```
In[2]:= url[y, z] = 3 qx[z] / h[z] × F0[(y - η[z]) / h[z]];
        wrl[y, z] = 3 qz[z] / h[z] × F0[(y - η[z]) / h[z]];
        vrl[y, z] = Integrate[D[wrl[y, z], z] /. y → yy, {yy, y, η[z]}];
```

Depth-integration of the in-plane momentum conservation equations

Ct=cotθ

```
In[5]:= eqrlx = 2 h[z]^3 / 3 == -ExpandAll[h[z]^2 / 3 ×
              Integrate[D[url[y, z], {y, 2}] + D[url[y, z], {z, 2}], {y, η[z], h[z] + η[z]}]]
        eqrlz = h[z]^3 / 3 (-2 Ct D[h[z] + η[z], z] + W D[κ[z], z]) == -ExpandAll[h[z]^2 / 3 ×
              Integrate[D[wrl[y, z], {y, 2}] + D[wrl[y, z], {z, 2}], {y, η[z], h[z] + η[z]}]]
```

Out[5]=
$$\frac{2\,h[z]^3}{3} == qx[z] - qx[z]\,h'[z]^2 + h[z]\,h'[z]\,qx'[z] - qx[z]\,h'[z]\,\eta'[z] + h[z]\,qx'[z]\,\eta'[z] + qx[z]\,\eta'[z]^2 + \frac{1}{2}\,h[z] \times qx[z]\,h''[z] - \frac{1}{3}\,h[z]^2\,qx''[z] + \frac{1}{2}\,h[z] \times qx[z]\,\eta''[z]$$

Out[6]=
$$\frac{1}{3}\,h[z]^3\,(-2\,Ct\,(h'[z] + \eta'[z]) + W\,\kappa'[z]) == qz[z] - qz[z]\,h'[z]^2 + h[z]\,h'[z]\,qz'[z] - qz[z]\,h'[z]\,\eta'[z] + h[z]\,qz'[z]\,\eta'[z] + qz[z]\,\eta'[z]^2 + \frac{1}{2}\,h[z] \times qz[z]\,h''[z] - \frac{1}{3}\,h[z]^2\,qz''[z] + \frac{1}{2}\,h[z] \times qz[z]\,\eta''[z]$$

Streamwise shear rate at the substrate

In[7]:=
```
taurl = D[url[y, z], y] /. y → η[z]
```
Out[7]=

$$\frac{3\,qx[z]}{h[z]^2}$$

§2.4.2 Weighted-residual integral boundary-layer model

Additional polynomial functions

In[8]:=
```
F1[zeta_] := zeta - 17 / 6 zeta^2 + 7 / 3 zeta^3 - 7 / 12 zeta^4
F2[zeta_] :=
 zeta - 13 / 2 zeta^2 + 57 / 4 zeta^3 - 111 / 8 zeta^4 + 99 / 16 zeta^5 - 33 / 32 zeta^6
```

Corrected velocity wall-normal profile ansatz

The leading-order velocity field, on which alone will act the second-order operators, e.g. d^2/dz^2, are separated into variables with the suffix 0. They are identical to the revised-lubrication ansatz above.

In[10]:=
```
uwr0[y, z] = 3 qx[z] / h[z] × F0[(y - η[z]) / h[z]];
wwr0[y, z] = 3 qz[z] / h[z] × F0[(y - η[z]) / h[z]];
vwr0[y, z] = Integrate[D[wwr0[y, z], z] /. y → yy, {yy, y, η[z]}];
uwr[y, z] = 3 (qx[z] - rx - sx) / h[z] F0[(y - η[z]) / h[z]] +
   45 rx / h[z] F1[(y - η[z]) / h[z]] + 210 sx / h[z] F2[(y - η[z]) / h[z]];
wwr[y, z] = 3 (qz[z] - rz - sz) / h[z] F0[(y - η[z]) / h[z]] +
    45 rz / h[z] F1[(y - η[z]) / h[z]] + 210 sz / h[z] F2[(y - η[z]) / h[z]];
vwr[y, z] = Integrate[D[wwr[y, z], z] /. y → yy, {yy, y, η[z]}];
```

Dynamic boundary conditions

In[15]:=
```
Dyu[z] = (D[h[z] + η[z], z] × D[uwr0[y, z], z]) /. {y → (h[z] + η[z])};
Dyw[z] =
  (4 D[h[z] + η[z], z] × D[wwr0[y, z], z] - D[vwr0[y, z], z]) /. {y → (h[z] + η[z])};
```

Pressure field

Ct=cotθ

In[17]:=
```
p[y, z] = 2 Ct (h[z] + η[z] - y) - W D[h[z] + η[z], {z, 2}] -
    D[wwr0[y, z], z] - (D[wwr0[y, z], z] /. {y → (h[z] + η[z])});
```

Streamwise Galerkin projections

In[18]:=
```
Galerkin0x =
  Integrate[F0[(y - η[z]) / h[z]] (2 + D[uwr0[y, z], {z, 2}]) + D[F0[(y - η[z]) / h[z]],
      {y, 2}] × uwr[y, z], {y, η[z], h[z] + η[z]}] + F0[1] × Dyu[z];
Galerkin1x =
  Integrate[F1[(y - η[z]) / h[z]] (2 + D[uwr0[y, z], {z, 2}]) + D[F1[(y - η[z]) / h[z]],
      {y, 2}] × uwr[y, z], {y, η[z], h[z] + η[z]}] + F1[1] × Dyu[z];
Galerkin2x = Integrate[F2[(y - η[z]) / h[z]] (2 + D[uwr0[y, z], {z, 2}]) +
     D[F2[(y - η[z]) / h[z]], {y, 2}] × uwr[y, z], {y, η[z], h[z] + η[z]}] + F2[1] × Dyu[z];
```

Transverse Galerkin projections

In[21]:=
```
Galerkin0z = Integrate[F0[(y - η[z]) / h[z]] (-D[p[y, z], z] + D[wwr0[y, z], {z, 2}]) +
     D[F0[(y - η[z]) / h[z]], {y, 2}] × wwr[y, z],
    {y, η[z], h[z] + η[z]}] + F0[1] × Dyw[z];
Galerkin1z = Integrate[F1[(y - η[z]) / h[z]] (-D[p[y, z], z] + D[wwr0[y, z], {z, 2}]) +
     D[F1[(y - η[z]) / h[z]], {y, 2}] × wwr[y, z],
    {y, η[z], h[z] + η[z]}] + F1[1] × Dyw[z];
Galerkin2z = Integrate[F2[(y - η[z]) / h[z]] (-D[p[y, z], z] + D[wwr0[y, z], {z, 2}]) +
     D[F2[(y - η[z]) / h[z]], {y, 2}] × wwr[y, z], {y, η[z], h[z] + η[z]}] + F2[1] × Dyw[z];
```

Momentum conservation equations

In[24]:=
```
eqwrx = 2 h[z]^3 / 3 == FullSimplify[2 h[z]^3 / 3 - h[z]^2 Galerkin0x]
```

Out[24]=

$$\frac{2\,h[z]^3}{3} == \frac{2}{5}\,h[z]\,\left(h'[z]\,qx'[z] - h[z]\,qx''[z]\right) + qx[z]\,\left(1 - \frac{3}{10}\,h'[z]^2 + \frac{1}{2}\,h'[z]\,\eta'[z] + \eta'[z]^2 + \frac{23}{40}\,h[z]\,h''[z] + \frac{3}{8}\,h[z]\,\eta''[z]\right)$$

In[25]:=
```
eqwrz = h[z]^3 / 3 (-2 Ct D[h[z] + η[z], z] + W D[h[z] + η[z], {z, 3}]) == FullSimplify[
    h[z]^3 / 3 (-2 Ct D[h[z] + η[z], z] + W D[h[z] + η[z], {z, 3}]) - h[z]^2 Galerkin0z]
```

Out[25]=

$$\frac{1}{3}\,h[z]^3\,\left(-2\,Ct\,\left(h'[z] + \eta'[z]\right) + W\,\left(h^{(3)}[z] + \eta^{(3)}[z]\right)\right) == \frac{9}{5}\,h[z]\,\left(h'[z]\,qz'[z] - h[z]\,qz''[z]\right) + qz[z]\,\left(1 - \frac{8}{5}\,h'[z]^2 + h'[z]\,\eta'[z] + 2\,\eta'[z]^2 + \frac{3}{10}\,h[z]\,\left(8\,h''[z] + 5\,\eta''[z]\right)\right)$$

r and s expressions

In[26]:=
```
rxslave = FullSimplify[(Last[List @@ eqwrx] - First[List @@ eqwrx]) 3 × 320 / 40 320 / 2 + rx /.
    Flatten[Solve[{Galerkin1x == 0, Galerkin2x == 0}, {rx, sx}]][[1]]]
sxslave =
 FullSimplify[sx /. Flatten[Solve[{Galerkin1x == 0, Galerkin2x == 0}, {rx, sx}]][[2]]]
```

Out[26]=
$$\frac{1}{13440}\left(-6\,h[z]\,qx'[z]\,(69\,h'[z]+133\,\eta'[z])-64\,h[z]^2\,qx''[z]+3\,qx[z]\left(206\,h'[z]^2+462\,h'[z]\,\eta'[z]+h[z]\,(15\,h''[z]-49\,\eta''[z])\right)\right)$$

Out[27]=
$$\frac{1}{1920}(14\,h[z]\,qx'[z]\,(h'[z]+\eta'[z])+qx[z]\,(-18\,h'[z]\,(h'[z]+\eta'[z])+h[z]\,(h''[z]+\eta''[z])))$$

In[28]:=
```
rzslave = FullSimplify[(Last[List @@ eqwrz] - First[List @@ eqwrz]) 3 × 320 / 40 320 / 2 + rz /.
    Flatten[Solve[{Galerkin1z == 0, Galerkin2z == 0}, {rz, sz}]][[1]]]
szslave =
 FullSimplify[sz /. Flatten[Solve[{Galerkin1z == 0, Galerkin2z == 0}, {rz, sz}]][[2]]]
```

Out[28]=
$$\frac{1}{6720}\left(-6\,h[z]\,qz'[z]\,(69\,h'[z]+133\,\eta'[z])-232\,h[z]^2\,qz''[z]+3\,qz[z]\left(206\,h'[z]^2+462\,h'[z]\,\eta'[z]+h[z]\,(99\,h''[z]+35\,\eta''[z])\right)\right)$$

Out[29]=
$$\frac{1}{960}(2\,h[z]\,(7\,qz'[z]\,(h'[z]+\eta'[z])+2\,h[z]\,qz''[z])-qz[z]\,(18\,h'[z]\,(h'[z]+\eta'[z])+5\,h[z]\,(h''[z]+\eta''[z])))$$

Streamwise shear rate at the substrate

In[30]:=
```
tauwr = ExpandAll[D[uwr[y, z], y] /. {y → η[z], rx → rxslave, sx → sxslave}]
```

Out[30]=
$$\frac{3\,qx[z]}{h[z]^2}-\frac{3\,qx[z]\,h'[z]^2}{320\,h[z]^2}+\frac{69\,h'[z]\,qx'[z]}{320\,h[z]}+\frac{153\,qx[z]\,h'[z]\,\eta'[z]}{64\,h[z]^2}-\frac{63\,qx'[z]\,\eta'[z]}{64\,h[z]}+\frac{159\,qx[z]\,h''[z]}{640\,h[z]}-\frac{qx''[z]}{5}-\frac{45\,qx[z]\,\eta''[z]}{128\,h[z]}$$